# $Id: PKGBUILD 184249 2013-05-05 00:06:28Z bisson $
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: damir <damir@archlinux.org>
# Contributor: Gan Lu <rhythm.gan@gmail.com>

pkgname=scim
pkgver=1.4.14
pkgrel=2
pkgdesc='Input method user interface and development platform'
url='http://www.scim-im.org/projects/scim'
license=('LGPL')
options=('!libtool')
arch=('i686' 'x86_64')
depends=('libltdl' 'gtk2')
makedepends=('intltool' 'optipng')
backup=('etc/scim/config' 'etc/scim/global')
source=("http://downloads.sourceforge.net/project/${pkgname}/${pkgname}/${pkgver}/${pkgname}-${pkgver}.tar.gz")
sha1sums=('73391be371dd7ef9f9af8b3a211e2bedec0cf71b')

install=install

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	./configure \
		--prefix=/usr \
		--sysconfdir=/etc \
		--disable-static \

	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
	find "${pkgdir}" -name *.png -exec optipng {} \; # libpng-1.6 has higher standards
}
